THE DEFICIENCY MODULE OF A CURVE AND ITS SUBMODULES

GUNTRAM HAINKE AND ALMAR KAID

1. Introduction

This is a summary of the fifth tutorial handed out at the CoCoA summer school 2005. We discuss the so-called deficiency module for projective curves. In particular, we provide the necessary CoCoA (version 4.6) codes for implementation. We thank Holger Brenner, Martin Kreuzer and Juan Migliore for helpful discussions.

Let K be a field and $P := K[x_0, \ldots, x_n]$ be a polynomial algebra equipped with the standard grading. By $\mathfrak{m} := (x_0, \ldots, x_n)$ we denote the graded maximal ideal. Moreover, by $I_C \subset P$ we denote a homogeneous ideal defining a curve $C \subset \mathbb{P}^n = \mathbb{P}^n_K = \operatorname{Proj} P$ and let $R := P/I_C$ be the coordinate ring of C. Throughout this article we assume that all ideals defining a projective subscheme are saturated, i.e.

$$I = I^{\text{sat}} = \{f \in P | \mathfrak{m}^t \cdot f \subset I \text{ for some } t > 0\}.$$

This can be checked with CoCoA, for example, with the function

```
Define IsSaturated(I)
  M:=Ideal(Indets());
Return Saturation(I,M)=I;
EndDefine;
```

Definition 1.1. The *deficiency module* or *Hartshorne-Rao module* of a curve $C \subset \mathbb{P}^n$ is the graded P-module

$$M(C) := \bigoplus_{t \in \mathbb{Z}} H^1(\mathbb{P}^n, \mathcal{I}_C(t)),$$

where $\mathcal{I}_C = \widetilde{I_C}$ is the ideal sheaf corresponding to I_C.

The deficiency module is also describable as the first local cohomology of $R = P/I_C$.

Proposition 1.2. *Let $C \subset \mathbb{P}^n$, $n \geq 2$, be a curve with coordinate ring $R = P/I_C$. Then $M(C) = H^1_{\mathfrak{m}}(R)$.*

Proof. From the short exact sequence

$$0 \longrightarrow I_C \longrightarrow P \longrightarrow R \longrightarrow 0$$

we derive the exact sequence in cohomology

$$H^1_{\mathfrak{m}}(P) \longrightarrow H^1_{\mathfrak{m}}(R) \longrightarrow H^2_{\mathfrak{m}}(I_C) \longrightarrow H^2_{\mathfrak{m}}(P).$$

Since $n \geq 2$, we have $H^1_{\mathfrak{m}}(P) = 0 = H^2_{\mathfrak{m}}(P)$ (cf. [2, Corollary A1.6]). Therefore, $H^1_{\mathfrak{m}}(R) \cong H^2_{\mathfrak{m}}(I_C)$. Now the assertion follows, since

$$H^2_{\mathfrak{m}}(I_C) = \bigoplus_{t \in \mathbb{Z}} H^1(\mathbb{P}^n, \mathcal{I}_C(t))$$

(cf. [2, Corollary A1.12(2)]). □

Definition 1.3. A curve $C \subset \mathbb{P}^n$ is called *arithmetically Cohen-Macaulay* if its coordinate ring $R = P/I_C$ is Cohen-Macaulay, i.e. $\dim R = \operatorname{depth} R$.

We have the following characterization for arithmetic Cohen-Macaulay curves utilizing its deficiency module.

Corollary 1.4. *Let $C \subset \mathbb{P}^n$, $n \geq 2$, be a curve. Then C is arithmetically Cohen-Macaulay if and only if $M(C) = 0$.*

Proof. Let C be arithmetically Cohen-Macaulay. Then we have that $H^i_{\mathfrak{m}}(R) = 0$ for all $i < 2$, since $\dim R = \operatorname{depth} R = 2$. Hence by Proposition 1.2 we have $M(C) = H^1_{\mathfrak{m}}(R) = 0$. On the other hand, supppose that $M(C) = H^1_{\mathfrak{m}}(R) = 0$. Because I_C is saturated, $H^0_{\mathfrak{m}}(R) = 0$. Since $\dim R = 2$, we have $H^2_{\mathfrak{m}}(R) \neq 0$ and $H^i_{\mathfrak{m}}(R) = 0$ for all $i > 2$. Hence $\dim R = \operatorname{depth} R$, i.e. R is Cohen-Macaulay and therefore C is arithmetically Cohen-Macaulay. (For all the vanishing and non-vanishing statements cf. [2, Proposition A1.16].) □

By using Proposition 1.2 above we can express the deficiency module of a curve in terms of Ext-modules.

Theorem 1.5. *Let $C \subset \mathbb{P}^n$, $n \geq 2$, be a curve with coordinate ring R. Then the modules $M(C)$ and $\operatorname{Hom}_K(\operatorname{Ext}^n_P(R, P), K)(n+1)$ are isomorphic as graded P-modules.*

Proof. By Proposition 1.2, we have $M(C) \cong H^1_{\mathfrak{m}}(R)$. The local duality Theorem (cf. [2, Theorem A1.9]) yields

$$H^1_{\mathfrak{m}}(R) \cong \operatorname{Ext}^n_P(R, P(-n-1))^* = \operatorname{Hom}_K(\operatorname{Ext}^n_P(R, P), K)(n+1)$$

and proves the claim. □

Theorem 1.5 allows a quick computation of the K-dual of $M(C)$ with CoCoA, using the implemented Ext-package. This is convenient for most of our computations in this paper. Therefore, we define the CoCoA-function `DeficiencyDual(I)` which takes the vanishing ideal I of a curve C and computes a presentation of the Ext-module $\mathrm{Ext}^n_P(R, P)$, where R is the coordinate ring of C:

```
Define DeficiencyDual(I)
  N:=NumIndets()-1;
Return Ext(N,CurrentRing()/I,Ideal(1));
EndDefine;
```

We now turn to finding a presentation of the K-dual of a P-module M which is a finite-dimensional $K-$vector space. Here we follow essentially the explanations in [1]. So let $M \cong P^k/N$ be a presentation of M and $<$ a module term ordering on P^k. Then a $K-$vector space basis of M is given by

$$B := \mathbb{T}^{n+1}\langle e_1, \dots, e_k\rangle \setminus \mathrm{LT}_<(N),$$

where $\mathbb{T}^{n+1}\langle e_1, \dots, e_k\rangle$ denotes the set of terms in $K[x_0, \dots, x_n]^k$ and $\mathrm{LT}_<(N)$ the leading term module of N with respect to $<$. For $te_i \in B$, let $\varphi_{t,i}$ denote the dual K-linear map, i.e. $\varphi_{t,i}(te_i) = 1$ and $\varphi_{t,i}(v) = 0$ for all $v \in B, v \neq te_i$. By definition of the P-linear structure on M^*, we have $f \cdot \varphi : m \mapsto \varphi(f \cdot m)$ for $f \in P, \varphi \in \mathrm{Hom}_K(M, K)$ and $m \in M$. This implies that $x_j \cdot \varphi_{t,i} = 0$ if $x_j \nmid t$ and $x_j \cdot \varphi_{t,i} = \varphi_{t',i}$ if $x_j t' = t$. In particular, a minimal system of generators for the K-dual M^* as a $P-$module is given by

$$E := \{\varphi_{t,i} : x_j \cdot te_i \in \mathrm{LT}_<(N) \text{ for } j = 0, \dots, n\}.$$

Clearly, there are two kinds of relations we have to take into account. The syzygies involving only one element correspond to the annihilator

$$\mathrm{Ann}_P\, \varphi_{t,i} = \langle x_0^{b_0}, \cdots, x_n^{b_n}\rangle$$

where $b_i = \deg_{x_i}(t) + 1$. The syzygies involving two elements are generated by those of the form

$$\frac{t}{\gcd(t,t')}\varphi_{t,i} - \frac{t'}{\gcd(t,t')}\varphi_{t',j} = 0$$

where $i = j$. Similar to the reasoning in [1, Proposition 5.3], it can be shown that these syzygies already generate the syzygy module. We sum up this discussion in the following proposition.

Proposition 1.6. *Let $M \cong P^k/N$ be a finite-dimensional $K-$vector space, $<$ a module term ordering on P^k and $E := \{t_\lambda \cdot e_{i(\lambda)}, \lambda \in \Lambda\}$ a monomial basis for M. Delete the subset $\{v \in E : x_j \cdot v \in E \text{ for some } j\}$ to obtain $E' = \{v_\lambda, \lambda \in \Gamma\}$, $\Gamma \subseteq \Lambda$. Let $N' \subseteq P^r$ denote the submodule generated by*

$$x_j^{b_{j,\lambda}} v_\lambda,\ j = 0, \dots, n, \lambda \in \Gamma$$

where $v_\lambda = t_\lambda e_{i(\lambda)}$ and $b_{j,\lambda} = \deg_{x_j}(t_\lambda) + 1$ and

$$\frac{t_\lambda}{\gcd(t_\lambda, t_\gamma)} v_\lambda - \frac{t_\gamma}{\gcd(t_\lambda, t_\gamma)} v_\gamma$$

where $i(\lambda) = i(\gamma)$, $\lambda, \gamma \in \Gamma$. Then there is a presentation

$$\mathrm{Hom}_K(M, K) \cong P^r/N'.$$

The implementation in CoCoA of this is somewhat lengthy. We include it here because the procedures `NormalBasis` and `SocleProj` which compute a $K-$vector space basis and the minimal generators come in handy in other instances too.

```
Define NormalBasis(M,Coord)
  G:=Gens(LT(M));
  N:=Len(G[1]);
  G:=[Vector(Q) | Q In G];
  NBasis:=[];
  L:=[Comp(List(G[J]),Coord) | J In 1..Len(G)];
  IList:=QuotientBasis(Ideal(L));
  NBasis:=Concat(NBasis,[Q*E_(Coord,N) | Q In IList]);
Return NBasis;
EndDefine;

Define NormalBasisM(M)
  G:=Gens(M);
  Nbr:=Len(G[1]);
Return ConcatLists([NormalBasis(M,I) | I In 1..Nbr]);
EndDefine;

Define SocleProj(M,Coord)
  NB:=NormalBasis(M,Coord);
  For I:=0 To NumIndets()-1 Do
    NB:=[Q In NB | Not(IsIn(x[I]*Q,NB))];
  EndFor;
```

```
Return NB;
EndDefine;

Define SocleProjM(M)
  G:=Gens(M);
  Nbr:=Len(G[1]);
Return ConcatLists([SocleProj(M,I) | I In 1..Nbr]);
EndDefine;

Define KDual(M)
  T:=SocleProjM(M);
  Nbr:=Len(T);
  N:=NumIndets()-1;
  Syz1:=[];
  For I:=1 To Nbr Do
    Syz1:=Concat(Syz1,[x[A]^(Deg(T[I],
          x[A])+1)*E_(I,Nbr) | A In 0..N]);
  EndFor;
  Syz2:=[];
  For I:=1 To Nbr-1 Do
    Pos:=FirstNonZeroPos(T[I]);
    For J:=I+1 To Nbr Do
      If FirstNonZeroPos(T[J])=Pos Then
        E1:=FirstNonZero(T[I]); E2:=FirstNonZero(T[J]);
        G:=GCD(E1,E2); RIJ:=E1/G; RJI:=E2/G;
        Syz2:=Concat(Syz2,[RIJ*E_(I,Nbr)-RJI*E_(J,Nbr)]);
      EndIf;
    EndFor;
  EndFor;
  Syz:=Concat(Syz1,Syz2);
Return(Module(Syz));
EndDefine;
```

In a special case, the following proposition allows an alternative way to obtain the deficiency module.

Proposition 1.7. *Let $C \subset \mathbb{P}^n$ be a curve which is the disjoint union of two components C_1 and C_2 which are both arithmetically Cohen-Macaulay with vanishing ideals I_{C_1} and I_{C_2} respectively. Then $M(C) = P/(I_{C_1} + I_{C_2})$ as a graded P-module.*

Proof. Let $X := \operatorname{Spec} R$ be the cone of C, let $X_i = \operatorname{Spec} P/I_{C_i}$, and let U, U_i be the corresponding punctured schemes (without the vertex), $i = 1, 2$. We consider the short exact sequence

$$0 \longrightarrow \Gamma(X, \mathcal{O}_X) \longrightarrow \Gamma(U, \mathcal{O}_X) \longrightarrow H^1_{\mathfrak{m}}(R) \longrightarrow 0$$

which combines sheaf cohomology and local cohomology (cf. [4, Exercise 2.3(e)]). Since $I_{C_1} \cap I_{C_2} = I_C$ we have the short exact sequence

$$0 \longrightarrow R = P/I_C \longrightarrow P/I_{C_1} \oplus P/I_{C_2} \longrightarrow P/(I_{C_1} + I_{C_2}) \longrightarrow 0.$$

Since $\Gamma(X, \mathcal{O}_X) = R = P/I_C$ and by assumption

$$\Gamma(U, \mathcal{O}_X) = \Gamma(U_1, \mathcal{O}_{X_1}) \oplus \Gamma(U_2, \mathcal{O}_{X_2}) = P/I_{C_1} \oplus P/I_{C_2}$$

holds, we get $M(C) = P/(I_{C_1} + I_{C_2})$. □

2. Examples for deficiency modules of curves

In the following we provide some examples of computations of deficiency modules via CoCoA. For this purpose we provide the useful function `GenLinForm(N)`. This function approximates a general linear form, i.e. it produces a linear form with randomized integer coefficients in the interval $[-N, N]$. This is sometimes computationally more convenient than `Randomized(DensePoly(1))`:

```
Define GenLinForm(N);
Return Sum([Rand(-N,N)*L | L In Indets()]);
EndDefine;
```

Example 2.1. Firstly, we consider the curve C_1 given as the union of two skew lines in $\mathbb{P}^3$.

We realize the situation in CoCoA via the commands

```
Use P::=Q[x[0..3]];
IL_1:=Ideal([GenLinForm(10),GenLinForm(10)]);
IL_2:=Ideal([GenLinForm(10),GenLinForm(10)]);
```

Now we apply Proposition 1.7 and set

```
I:=IL_1+IL_2;
```

and compute the deficiency module of C_1 via

```
MC_1:=P/I;
```

By using the command

```
Hilbert(MC_1);
```

we obtain the Hilbert function of $M(C_1)$ and get the CoCoA answer:

```
H(0) = 1
H(t) = 0   for t >= 1
-------------------------------
```

i.e. $M(C_1)_0 = K$ and $M(C_1)_t = 0$ for $t > 0$. In particular C_1 is not arithmetically Cohen-Macaulay.

Example 2.2. Let $C_2 := V_+(x_0x_2 - x_1^2, x_1x_3 - x_2^2, x_0x_3 - x_1x_2) \subset \mathbb{P}^3$ be the twisted cubic curve. So we use the commands

```
Use P::=Q[x[0..3]];
IC_2:=Ideal(x[0]x[2]-x[1]^2,x[1]x[3]-x[2]^2,
      x[0]x[3]-x[1]x[2]);
```

and compute the resolution of P/I_{C_2} by

```
Res(P/IC_2);
```

CoCoA yields:

```
0 --> P^2(-3) --> P^3(-2) --> P
-------------------------------
```

We see that $\operatorname{pd}(P/I_{C_2}) = 2 = \operatorname{codim} I_{C_2}$ where $\operatorname{codim} I = \min\{\operatorname{ht}(\mathfrak{p}) : I \subseteq \mathfrak{p} \text{ minimal}\}$. Hence $R = P/I_{C_2}$ is Cohen-Macaulay and therefore by Corollary 1.4 we get $M(C_2) = 0$.

Example 2.3. Here we consider the smooth rational quartic curve $C_3 \subset \mathbb{P}^3$ defined as the vanishing of the 2×2 minors of the matrix

$$\begin{pmatrix} x_0 & x_1^2 & x_1x_3 & x_2 \\ x_1 & x_0x_2 & x_2^2 & x_3 \end{pmatrix}.$$

We realize the vanishing ideal of the curve C_3 by the commands

```
Use P::=Q[x[0..3]];
M:=Mat([[x[0],x[1]^2,x[1]x[3],x[2]],
        [x[1],x[0]x[2],x[2]^2,x[3]]]);
IC_3:=Ideal(Minors(2,M));
```

Now we look at the K-dual of the deficiency module $M(C_3)$ by

```
MC_3:=DeficiencyDual(IC_3);
```

If we compute the Hilbert function of $M(C_3)$ via

```
Hilbert(MC_3);
```

we get

```
H(0) = 1
H(t) = 0   for t >= 1
-------------------------------
```

Hence $M(C_3)$ is one-dimensional.

Example 2.4. In this example we want to study the deficiency module of a curve $C_4 \subset \mathbb{P}^3$ given as the disjoint union of a line and a plane curve of degree d for some $d \in \mathbb{N}$.

So we realize the situation (for $d = 3$) with CoCoA in the following way:

```
Use P::=Q[x[0..3]];
IL:=Ideal([GenLinForm(10),GenLinForm(10)]);
Use S::=Q[x[0..2]];
D:=3;
F:=Randomized(DensePoly(D));
Use P;
IPC:=Ideal(BringIn(F),x[3]);
```

Now we apply again Proposition 1.7 and compute the deficiency module $M(C_4)$ via

```
I:=IL+IPC;
MC_4:=P/I;
```

Now we compute the Hilbert function of $M(C_4)$ by

```
Hilbert(MC_4);
```

and get

```
H(0) = 1
H(1) = 1
H(2) = 1
H(t) = 0   for t >= 3
-------------------------------
```

i.e. in the case $d = 3$ we have $\dim_K(M(C_4)) = 3$. If we do this for various values d we get $\dim_K(M(C_4)) = d$ as it should be.

Example 2.5. We want to compute the deficiency module for a curve $C_5 \subset \mathbb{P}^3$ given as the union of a line and a plane curve of degree d, $d \in \mathbb{N}$, which meet in (at least) one point.

To do this we fix the (intersection) point $y := V_+(x_1, x_2, x_3)$. So we compute the vanishing ideal of C_5 (for $d = 3$) in the following way:

```
Use P::=Q[x[0..3]];
G_1:=Sum([Rand(-10,10)*x[T]|T In 1..3]);
G_2:=Sum([Rand(-10,10)*x[T]|T In 1..3]);
IL:=Ideal(G_1,G_2);
```

So I_L is the ideal corresponding to a line in $\mathbb{P}^3$ passing through y. Now we construct an appropriate plane curve meeting the line L in the point y.

```
Use S::=Q[x[0..2]];
D:=3;
F:=Randomized(DensePoly(D)-x[0]^D);
Use P;
IPC:=Ideal(BringIn(F),x[3]);
```

Hence we get the vanishing ideal of C_5 by

```
IC_5:=Intersection(IL,IPC);
```

and obtain the K-dual of the deficiency module as usual by

```
DeficiencyDual(IC_5);
```

CoCoA yields:

```
Module([0])
-------------------------------
```

As in the previous example we get the same result for various values of d. So it is arithmetically Cohen-Macaulay.

Example 2.6. We consider the coordinate cross in $\mathbb{P}^3$ defined by the intersection I_{cross} of the ideals (x_1, x_2), (x_1, x_3) and (x_2, x_3) in $P = K[x_0, x_1, x_2, x_3]$. We realize I in CoCoA via

```
Use P::=Q[x[0..3]];
L:=[Ideal(x[1],x[2]),Ideal(x[1],x[3]),Ideal(x[2],x[3])];
ICross:=IntersectionList(L);
```

Now we compute a regular sequence of type $(3,3)$ in I_{cross} which defines a complete intersection curve containing the coordinate cross. To do this we use the function `GenRegSeq(I,L)` defined in [3]:

```
ICi:=Ideal(GenRegSeq(ICross,[3,3]));
```

Now we look at the vanishing ideal I_X of the residual curve X by

```
IX:=ICi:ICross;
```

If we compute the Hilbert polynomial of X via

```
Hilbert(P/IX);
```

we get $\mathrm{HP}_X(t) = \mathrm{HP}_{P/I_X}(t) = 6t - 2$ for $t \geq 1$, i.e. X is a curve of degree 6 and genus 3. Furthermore, by using

```
DeficiencyDual(IX);
```

we notice that $M(X) = 0$, i.e. by Corollary 1.4 the curve X is arithmetically Cohen-Macaulay. We check with CoCoA that X is a smooth curve by computing the singular locus:

```
Define IsSmooth(I)
  J:=Jacobian(Gens(I));
  L:=List(Minors(1,J));
  SingLoc:=Radical(Ideal(L)+I);
Return SingLoc=Ideal(Indets());
EndDefine;

IsSmooth(IX);
```

Next, we construct another smooth curve in $\mathbb{P}^3$ of degree 6 and genus 3 which is not arithmetically Cohen-Macaulay. For this we go back to Example 2.1 and consider the curve C_1 given as the union of two skew lines in $\mathbb{P}^3$. We obtain the vanishing ideal of C_1 by

```
IC_1:=Intersection(IL_1,IL_2);
```

We have already shown that C_1 is not arithmetically Cohen-Macaulay and it is known that this is invariant under linkage. In the first step we link the curve C_1 via a complete intersection curve of type $(3, 4)$ to a curve X_1. To do this we use the commands:

```
ICi_1:=Ideal(GenRegSeq(IC_1,[3,4]));
IX_1:=ICi_1:IC_1;
```

Now we link the curve X_1 further via a complete intersection curve of type $(4, 4)$ to a curve X_2:

```
ICi_2:=Ideal(GenRegSeq(IX_1,[4,4]));
IX_2:=ICi_2:IX_1;
```

Now we check the Hilbert polynomial with

```
Hilbert(P/IX_2);
```

and get $\mathrm{HP}_{X_2}(t) = \mathrm{HP}_{P/I_{X_2}}(t) = 6t - 2$ for $t \geq 1$. And indeed, we use

```
DeficiencyDual(IX_2);
```

to check that $M(X_2) \neq 0$, i.e. X_2 is not arithmetically Cohen-Macaulay. Finally, the smoothness of X_2 is established via

```
IsSmooth(IX_2);
```

3. A degree bound for curves in $\mathbb{P}^3$

In this section we want to study a degree bound for curves in $\mathbb{P}^3$ given in [5] and compute some examples. In the sequel C denotes a curve in $\mathbb{P}^3$. Firstly, we define a submodule of the deficiency module $M(C)$ of C.

Definition 3.1. Let $\ell, \ell' \in P_1$ be two general linear forms and let $A := (\ell, \ell')$ denote the ideal they generate. Then we define the P-submodule $\mathcal{K}_A \subseteq M(C)$ as the submodule annihilated by A, i.e.

$$\mathcal{K}_A = 0 :_{M(C)} A.$$

We want to compute with CoCoA a presentation of the module $\mathcal{K}_A$. To do this, we need a presentation for the annihilator $K_\ell := 0 :_{M(C)} (\ell)$ first, where $\ell \in P_1$ is a general linear form. So we write a function `K_L(I)`, which computes a presentation of K_ℓ for a curve defined by the ideal I.

```
Define K_L(I)
  MCdual:=DeficiencyDual(I);
  MC:=KDual(MCdual);
  GensMC:=Gens(MC);
  N:=Len(GensMC[1]);
  F:=GenLinForm(10);
  L:=Concat([F*E_(I,N) | I In 1..N],GensMC);
  S:=Syz(L);
  GensCol:=[Vector(First(List(T),N)) | T In Gens(S)];
Return Module(GensCol);
EndDefine;
```

With the help of the function `K_L(I)` above we can easily write a function `K_A(I)`, which computes a presentation of the submodule $\mathcal{K}_A \subseteq M(C)$ for a curve defined by the ideal I:

```
Define K_A(I)
  KL_1:=K_L(I);
  KL_2:=K_L(I);
Return(Intersection(KL_1,KL_2));
EndDefine;
```

Definition 3.2. Let M be a finitely generated graded P-module with graded minimal free resolution

$$0 \longrightarrow \bigoplus_{j=1}^{b_s} P(-d_{s,j}) \longrightarrow \cdots \longrightarrow \bigoplus_{j=1}^{b_0} P(-d_{0,j}) \longrightarrow M \longrightarrow 0.$$

We define

$$m_i(M) := \min\{d_{i,1}, \ldots, d_{i,b_i}\}.$$

The following theorem (cf. [5, Corollary 2.3]), which provides a lower bound for the degree of a curve in $\mathbb{P}^3$, builds the fundament for our further studies.

Theorem 3.3. *Let $C \subset \mathbb{P}^3$ be a curve. If $m_2(P/I_C) \leq m_1(\mathcal{K}_A) + 2$ then*

$$\deg(C) \geq \frac{1}{2} \cdot m_1(P/I_C) \cdot m_2(P/I_C) - \dim_K(\mathcal{K}_A).$$

Example 3.4. We come back to Example 2.1 and consider the curve C_1 given by the union of two skew lines in $\mathbb{P}^3$.

Since we have already shown that $M(C_1) = K \neq 0$, we know that C_1 is not arithmetically Cohen-Macaulay (cf. Corollary 1.4). Next we compute the vanishing ideal of C_1 by

```
IC_1:=Intersection(IL_1,IL_2);
```

where `IL_1` and `IL_2` were defined in Example 2.1. Now we compute the Hilbert polynomial of the curve C_1 utilising the command

```
Hilbert(P/IC_1);
```

and get

```
H(0) = 1
H(t) = 2t + 2   for t >= 1
-------------------------------
```

i.e the Hilbert polynomial of C_1 equals $\mathrm{HP}_{C_1}(t) = 2t + 2$. Hence $\deg C_1 = 2$. Now we obtain the numbers $m_1(P/I_{C_1})$ and $m_2(P/I_{C_1})$ by computing the resolution of P/I_{C_1}:

```
Res(P/IC_1);
```

The answer of CoCoA is

```
0 --> P(-4) --> P^4(-3) --> P^4(-2) --> P
-------------------------------
```

Thus we get $m_1(P/I_{C_1}) = 2$ and $m_2(P/I_{C_1}) = 3$. This example illustrates that the stronger degree bound

$$\deg(C) \geq \frac{1}{2} \cdot m_1(P/I_C) \cdot m_2(P/I_C)$$

does not hold in general for curves which are not arithmetically Cohen-Macaulay, since for the curve C_1 the right hand side equals 3.

Example 3.5. We consider a curve $C \subset \mathbb{P}^3$ given as the disjoint union of two complete intersections of type $(16, 16)$ (cf. also [5, Example 2.7]).

To simplify matters we consider the complete intersections $I_1 := (x_0^{16}, x_1^{16})$ and $I_2 := (x_2^{16}, x_3^{16})$. We realise the vanishing ideal of C via

```
Use P::=Q[x[0..3]];
I_1:=Ideal(x[0]^16,x[1]^16);
I_2:=Ideal(x[2]^16,x[3]^16);
IC:=Intersection(I_1,I_2);
```

We use Proposition 1.7 to compute the deficiency module $M(C)$:

```
I:=I_1+I_2;
MC:=P/I;
```

Since $I_1 + I_2$ is an $\mathfrak{m}$-primary ideal, the Hilbert function of $M(C) = P/(I_1+I_2)$ has only finitely many non-zero values. Therefore, we obtain the dimension of $M(C)$ by

```
Sum:=0;
T:=0;
While Hilbert(MC,T)<>0 Do
  Sum:=Hilbert(MC,T)+Sum;
  T:=T+1;
EndWhile;
```

Now the command `Sum;` yields $\dim_K(M(C)) = 65536$. We compute the Hilbert polynomial of our curve C as usual by

```
Hilbert(P/IC);
```

and verify that the degree of C is $512 = 2 \cdot 16^2$ as expected. Next we compute the minimal graded free resolutions of P/I_C with

```
Res(P/IC);
```

and get

```
0 --> P(-64) --> P^4(-48) --> P^4(-32) --> P
-------------------------------
```

Hence $m_1(P/I_C) = 32$ and $m_2(P/I_C) = 48$. Now we compute a representation for the annihilator $\mathcal{K}_\ell$ for a general linear form ℓ. Since we have given the deficiency module $M(C) = P/I$ as a quotient of the polynomial ring with $I = I_1 + I_2$ in this special example, we can compute it without using the Ext-package in the following way: We have $K_\ell = \{(g + I) \in P/I : \ell \cdot (g + I) = 0 + I\}$. Therefore we have to compute the preimage of K_ℓ in the polynomial ring P which equals the ideal quotient $J_1 := I :_P (\ell)$. To do this we use the commands

```
L_1:=Ideal(GenLinForm(10));
J_1:=I:L_1;
```

To get the dimension of $\mathcal{K}_\ell$ as a K-vector space we compute at first via

```
Len(QuotientBasis(J_1));
```

the lenght of a K-basis of P/J_1 which equals 62800. Hence we have $\dim_K(\mathcal{K}_\ell) = 65536 - 62800 = 2736$. In the same way we compute a representation of the submodule $\mathcal{K}_A \subseteq M(C)$:

```
L_2:=Ideal(GenLinForm(10),GenLinForm(10));
J_2:=I:L_2;
```

The command

```
Len(QuotientBasis(J_2));
```

yields $\dim_K(P/J_2) = 65365$ and therefore we have

$$\dim_K(\mathcal{K}_A) = 65536 - 65365 = 171.$$

Next we want to compute the value $m_1(\mathcal{K}_A)$, i.e. the minimal degree of a generator of $\mathcal{K}_A$. To do this we compute the graded minimal free resolution of P/I and P/J_2 respectively by

```
Res(P/I);
Res(P/J_2);
```

and get

```
0 --> P(-64) --> P^4(-48) --> P^6(-32) --> P^4(-16) --> P
-------------------------------
0 --> P^2(-63) --> P(-42)(+)P^2(-43)(+)P^8(-47)(+)P(-62)
--> P^6(-32)(+)P^2(-41)(+)P^4(-42)(+)P^4(-46) -->
P^4(-16)(+)P(-40)(+)P^2(-41) --> P
-------------------------------
```

Since $\mathcal{K}_A \cong J_2/I$ we see that $m_1(\mathcal{K}_A) = 40$. Altogether we have

$$m_2(P/I_C) = 48 > 42 = 40 + 2 = m_1(\mathcal{K}_A) + 2$$

and

$$\begin{aligned} \deg C = 512 < 597 &= \frac{1}{2} \cdot 32 \cdot 48 - 171 \\ &= \frac{1}{2} \cdot m_1(P/I_C) \cdot m_2(P/I_C) - \dim_K(\mathcal{K}_A). \end{aligned}$$

Hence this example illustrates that the assumption

$$m_2(P/I_C) \leq m_1(\mathcal{K}_A) + 2$$

in Theorem 3.3 is necessary and can not be dropped.

Example 3.6. We take a (non-reduced) curve $C \subset \mathbb{P}^3$ defined by the ideal $I_C := (x_0, x_1)^{12} + (f)$, where $f \in (x_0, x_1)$ is a generic form of degree 15 (cf. also [5, Example 2.8]). This can be realized in CoCoA via the following instructions:

```
Use P::=Q[x[0..3]];
J:=Ideal(x[0],x[1])^12;
F:=x[0]*GenForm(14,50)+x[1]*GenForm(14,50);
IC:=J+Ideal(F);
```

Here the procedure `GenForm(D,N)` returns a general form of degree D with randomized coefficients in the interval $[-N, N]$:

```
Define GenForm(D,N)
  V:=Gens(Ideal(Indets())^D);
Return Sum([Rand(-N,N)*V[I] | I In 1..Len(V)]);
EndDefine;
```

If we compute the Hilbert polynomial of C with

```
Hilbert(P/IC);
```

we get

$$\mathrm{HP}_C(t) = \mathrm{HP}_{P/I_C}(t) = 12t + 870$$

for $t \geq 24$, i.e. $\deg(C) = 12$. Furthermore, the computation of the minimal graded free resolution of P/I_C via

```
Res(P/IC);
```

yields

```
0 --> P^11(-27) --> P^12(-13)(+)P^12(-26)
--> P^13(-12)(+)P(-15) --> P
-------------------------------
```

Hence $m_1(P/I_C) = 12$ and $m_2(P/I_C) = 13$. We obtain the deficiency module $M(C)$ and its submodule $\mathcal{K}_A$ via

```
MC:=KDual(DeficiencyDual(IC));
KA:=K_A(MC);
```

The dimension of the deficiency module can be calculated, for example, as the length of a normal basis defined in Section 1. The command

```
Len(NormalBasisM(MC));
```

then gives $\dim_K M(C) = 56056$. Likewise,

```
Len(NormalBasisM(KA));
```

gives the codimension of $\mathcal{K}_A$ in $M(C)$, which yields $\dim_K \mathcal{K}_A = 66$. The problem of determining $m_1(\mathcal{K}_A)$, i.e. the degree of a minimal generator of $\mathcal{K}_A$, is more delicate, since degree shifts are not supported in the current version of CoCoA. We work around this as follows: Looking at the resolution of I_C, we find that $\mathrm{Ext}^3(P/I_C, P) \cong P^{11}(27)/N$. Calculating the Hilbert function via

```
Hilbert(DeficiencyDual(IC));
```

we get nontrivial terms in degrees 0 to 166, i.e. $\operatorname{Ext}^3(P/I_C, P)$ is concentrated in degrees -27 to 139. The K-dual then is concentrated in degrees -139 to 27, and taking the degree shift into account as given in Theorem 1.5, we see that the last non-trivial component of the deficiency module is in degree 23. Calculating the Hilbert function of $\mathcal{K}_A$ with the help of `HilbertSeriesShifts` we find that the Hilbert function of `KA` is of the shape

$$\mathrm{HF}_{\mathcal{K}_A} : 11,\ 10,\ 9,\ 8,\ 7,\ 6,\ 5,\ 4,\ 3,\ 2,\ 1,\ 0.$$

Taking the various shifts and twists into account, this has to be interpreted as

$$\mathrm{HF}_{\mathcal{K}_A}(t) = t - 12$$

for $13 \leq t \leq 23$ and $\mathrm{HF}_{\mathcal{K}_A}(t) = 0$ otherwise. In particular, we get $m_1(\mathcal{K}_A) = 13$.

Hence the assumption

$$m_2(P/I_C) = 13 \leq 15 = 13 + 2 = m_1(\mathcal{K}_A) + 2$$

of Theorem 3.3 holds and the degree bound given there is sharp since we have

$$\deg(C) = 12 = \frac{1}{2} \cdot 12 \cdot 13 - 66 = \frac{1}{2} \cdot m_1(P/I_C) \cdot m_2(P/I_C) - \dim_K(\mathcal{K}_A).$$

References

[1] S. Beck and M. Kreuzer. How to compute the canonical module of a set of points. *Algorithms in algebraic geometry and applications (Santander, 1994)*, volume 143 of *Progr. Math.*, pages 51–78. Birkhäuser, Basel, 1996.

[2] D. Eisenbud. *The Geometry of Syzygies.* Springer-Verlag, 2005.

[3] G. Hainke and A. Kaid. Constructions of Gorenstein rings. *in this volume*, 2006.

[4] R. Hartshorne. *Algebraic Geometry.* Springer, New York, 1977.

[5] J. Migliore, U. Nagel, and T. Römer. Extensions of the multiplicity conjecture. *ArXiv*, 2005.

Fakultät für Mathematik, Universität Bielefeld, D-33501 Bielefeld, Germany

E-mail address: `ghainke@math.uni-bielefeld.de`

Department of Pure Mathematics, University of Sheffield, Hounsfield Road, Sheffield S3 7RH, United Kingdom

E-mail address: `A.Kaid@sheffield.ac.uk`